
// --- settings
//RSIPeriod=14
//Stochastic1PeriodK  = 8
//Stochastic1PeriodD  = 3
//Stochastic2PeriodK  = 14
//Stochastic2PeriodD  = 3
//DrawHullTrend = 1 //(1=true ; 0=false)
//HullPeriod     =12
//HullDivisor    =2
// --- end of settings

once threshold=50
once ilimit=0

bufRSI=rsi[RSIPeriod]
bufStoch1=stochastic[Stochastic1PeriodK,Stochastic1PeriodD]
bufStoch2=stochastic[Stochastic2PeriodK,Stochastic2PeriodD]
bufRSI=bufRSI-threshold
bufStoch1=bufStoch1-threshold
bufStoch2=bufStoch2-threshold

bufHistUP=0
bufHistDN=0
bufHistNO=0
if(bufRSI>ilimit and bufStoch1>ilimit and bufStoch2>ilimit) then
bufHistUP=bufStoch2
else
if(bufRSI<ilimit and bufStoch1<ilimit and bufStoch2<ilimit) then
bufHistDN=bufStoch2
else
bufHistNO=bufStoch2
endif
endif

if (DrawHullTrend) then
inner = 2*weightedaverage[ round( HullPeriod/HullDivisor ) ](close)-weightedaverage[HullPeriod](close)
MMHULL=weightedaverage[ round( sqrt(HullPeriod) ) ]( inner )
r=255
g=155
if MMHULL>MMHULL[1] then
r=0
g=255
endif
drawtext("●",barindex,0,dialog,bold,16) coloured(r,g,0)
endif

return bufHistUP coloured(0,128,0) style(histogram) as "Up", bufHistDN coloured(255,0,0) style(histogram) as "Dn", bufHistNO coloured(169,169,169) style(histogram) as "No",0 coloured(168,168,168) style(point,1) as "level0", 20  coloured(168,168,168) style(dottedline,1) as "level20", -20  coloured(168,168,168) style(dottedline,1) as "level-20"
